
1. Введение 

   Язык Z2 разработан для расчета начислений или удержаний по видам оплат и создания выходных печатных форм в системе БОСС-Кадровик (http://www.bosshr.ru). Тем не менее, он является универсальным языком программирования, на котором можно писать любые алгоритмы в рамках платформы RP Server.
   Его основные черты - блочная структура, статическая типизация, а также расширение типов и связанные с типами функции. 
  Синтаксис языковых конструкций базируется на языке Си. Концепция расширения структуры, связанных с типами функций и само описание заимствованы из языка Оберон-2. Расширение типов делает Z2 объектно-ориентированным языком. Объект - это переменная абстрактного типа, содержащая данные (состояние объекта) и функции, которые оперируют этими данными.

   Этот документ не является учебником программирования. Его назначение - служить справочником для программистов.

   В приложении A определены некоторые термины, которые используются при описании правил соответствия типов Z2. В тексте эти термины выделены курсивом, чтобы подчеркнуть их специальное значение (например, одинаковый тип). 

2. Синтаксис

Для описания синтаксиса Z2 используются Расширенные Бэкуса-Наура Формы (РБНФ). Варианты разделяются знаком |. Квадратные скобки [ и ] означают необязательность записанного внутри них выражения, а фигурные скобки { и } означают его повторение (возможно 0 раз). Нетерминальные символы начинаются с заглавной буквы (например, Оператор). Терминальные символы заключаются в кавычки (например, "="). 

3. Словарь и представление 

Для представления терминальных символов предусматривается использование набора знаков ASCII. Слова языка - это идентификаторы, числа, строки, операции и разделители. Должны соблюдаться следующие лексические правила. Пробелы и концы строк не должны встречаться внутри слов (исключая комментарии и пробелы в символьных строках). Пробелы и концы строк игнорируются, если они не существенны для отделения двух последовательных слов. Заглавные и строчные буквы считаются различными. 

1. Идентификаторы - последовательности букв и цифр и знаков подчеркивания. Первый символ должен быть буквой или знаком подчеркивания.

   идент = (буква | “_” {буква | цифра | “_” }. 

Примеры: x  pay  Z2  перваяБуква

2. Числа - вещественные  числа. Вещественное число может содержать десятичную точку. 

     Число  = цифра { цифра } [ "." { цифра }] .

Примеры:   1994, 3.14159

4. Строки - последовательности символов, заключенные в двойные (") кавычки. Число символов в строке называется ее длиной. 

   строка = ' " ' {символ} ' " '. 

Некоторые символы в строках записываются в виде последовательности “\символ”

\” - двойная кавычка

\n – новая строка 

\r – возврат каретки – равносилен символу новой строки

\t – горизонтальная табуляция

\\ - обратная наклонная черта (обратный слэш)

Примеры: "Все будет хорошо"    "АО \”Тополь\”"    "x".     В последнем примере видно, как использовать двойные кавычки в составе строки.

5. Операции и разделители - это специальные символы, пары символов или зарезервированные слова, перечисленные ниже. Зарезервированные слова состоят исключительно из строчных букв и не могут использоваться как идентификаторы. 

	+       
	=       
	           
	
	

	-
	
	break
	прервать
	

	*
	==
	const
	константа
	

	/
	!=
	continue
	продолжить
	

	!
	<
	if
	если
	

	&
	>
	return
	вернуть
	

	.
	<=
	struct
	структура
	 

	,
	>=
	sizeof
	размер
	

	;
	:
	else
	иначе
	

	||
	&&
	function
	функция
	

	(
	)
	while
	цикл  
	

	[
	]
	import
	импорт
	

	{
	}
	isstruct
	
	


Пары (break  прервать, if если и т.д.)  являются синонимами.

6. Комментарии могут быть вставлены между любыми двумя словами программы. Это произвольные последовательности символов, начинающиеся знаками /* и оканчивающиеся */. Комментарии могут быть вложенными. Они не влияют на смысл программы. Есть также однострочные комментарии, начинающиеся со знаков // и продолжающиеся до конца строки. Однострочные комментарии вида //@L+  или //@L- применяются для управления компиляцией. Здесь вместо L используется одна из букв латинского алфавита. В настоящее время задействованы только буквы B и D. (см. главу 4. и  ОператорыSQL). 

4. Объявления и области действия 

Каждый идентификатор, встречающийся в программе, должен быть объявлен, если это не стандартный идентификатор. Объявления задают некоторые постоянные свойства объекта, например, является ли он константой, типом, переменной или процедурой. После объявления идентификатор используется для ссылки на соответствующий объект. Числовые переменные можно не объявлять, если первым упоминанием переменной будет появление ее слева от знака присваивания. Однако, если установлена директива компилятора @D, то объявления обязательны и для числовых переменных.

Область действия объекта x распространяется текстуально от точки его объявления до конца блока (модуля, составного оператора, функции или структуры), в котором находится объявление. Для этого блока объект является локальным. Это разделяет области действия одинаково именованных объектов, которые объявлены во вложенных блоках. Правила для областей действия таковы: 

1. Идентификатор не может обозначать больше чем один объект внутри данной области действия (то есть один и тот же идентификатор не может быть объявлен в блоке дважды); 

2. Ссылаться на объект можно только изнутри его области действия; 

3. Идентификаторы, обозначающие поля структуры (см. 6.3) или функции, связанные с типом, (см. 10.2) могут употребляться только в обозначениях структуры.

Идентификатор, объявленный в блоке модуля, может сопровождаться при своем объявлении экспортной меткой "*", чтобы указать, что он экспортируется. Идентификатор x, экспортируемый модулем M, может использоваться в других модулях, если они импортируют M (см. гл. 11). Тогда идентификатор обозначается в этих модулях М.x и называется уточненным идентификатором. Экспортироваться могут только константы, структуры и их поля и функции. Переменные не могут экспортироваться.

УточнИдент =  [идент "."] идент.

ИдентОпр = идент [ "*" ].

Следующие идентификаторы являются стандартными; их значение определено в указанных разделах: 

	number   
	(6.1)
	число
	(6.1)

	var
	(6.1)
	перем
	(10.3)

	string
	(6.1)
	строка
	(6.1)

	isstruct
	(10.3)
	sizeof
	(10.3)

	strlen
	(10.3)
	strtrunc
	(10.3)

	strdel
	(10.3)
	strins
	(10.3)

	strcmp
	(10.3)
	strncmp
	(10.3)

	strcpy
	(10.3)
	strncpy
	(10.3)

	strcat
	(10.3)
	strncat
	(10.3)

	strstr
	(10.3)
	strnstr
	(10.3)

	strnum
	(10.3)
	numstr
	(10.3)

	topen
	(10.3)
	tclose
	(10.3)

	tget
	(10.3)
	teot
	(10.3)

	DO_SQL
	(10.3)
	
	(10.3)


5. Объявления констант 

Объявление константы связывает ее идентификатор с ее значением. 

"const" ИдентОпр "=" КонстВыраж { "," ИдентОпр "=" КонстВыраж } ";"

Константное выражение - это выражение, которое может быть вычислено по его тексту без фактического выполнения программы. Его операнды - константы (Гл. 8) или стандартные функции (Гл. 10.3), которые могут быть вычислены во время компиляции. 
Примеры объявлений констант: 

const N = 100,  limit = 2*N – 1;
const размер = sizeof(A); 

6. Объявления типа 

Тип данных определяет набор значений, которые могут принимать переменные этого типа, и набор применимых операций. Объявление типа связывает идентификатор с типом. В случае структурированных типов (массивы и записи) объявление также определяет структуру переменных этого типа. Структурированный тип не может содержать сам себя. 

Примеры: 

struct Point2 { number x, y; };

struct Point3 : Point2 { number z; };

6.1 Основные типы 

Основные типы обозначаются стандартными идентификаторами. Соответствующие операции определены в 8.2, а стандартные функции в 10.3. Предусмотрены следующие основные типы: 

	1.
	string                 
	символы расширенного набора ASCII (0X .. 0FFX)

	2.
	number                  
	вещественные числа


Как таковой тип string не существует. Он в действительности является последовательностью символов, заканчивающейся нулевым символом. Размер строки должен включать завершающий нулевой символ. Строки запоминаются в массивах типа string. Непосредственного доступа к символам строки нет.

6.2 Тип массив 

Массив – совокупность данных, состоящая из определенного количества элементов одинакового типа, называемого типом элементов. Число элементов массива называется его длиной. Элементы массива обозначаются индексами, которые являются целыми числами от 0 до длины массива минус 1. Тип массив неявно определяется при объявлении переменных.

Тип идент ArraySize  ";"

ArraySize = { "[" [ КонстВыраж ] "]" } .

Тип  =  УточнИдент .

Объявление  вида  T a[L0][L1] ...[Ln]  понимается как массив из L0 элементов вида T a[L1] ...[Ln].

Массивы, объявленные без указания длины, называются открытыми массивами. Они могут использоваться только в качестве типа формального параметра (см. 10.1). 

Для массивов –строк длина может быть опущена. Тогда она предполагается равной 256.

Т.е. описания string s[256]  и string s одинаковы. Длина строки включает завершающий нулевой символ. Следовательно,  в строке может храниться на один символ меньше указанного в объявлении.

Примеры: 

number a[10][N] ;
string  s[20][200], s2; 

6.3 Тип структура 

Тип структура – совокупность данных, состоящая из фиксированного числа элементов, которые могут быть различных типов и называются полями. Объявление типа структура определяет имя и тип каждого поля. Область действия идентификаторов полей простирается от точки их объявления до конца объявления типа запись, но они также видимы внутри обозначений, ссылающихся на элементы переменных-записей (см. 8.1). 

  "struct" ИдентОпр [ ":" Тип ] "{" { Тип ИдентОпр ArraySize { "," ИдентОпр ArraySize } ";" } "}" ";"

  ArraySize = [] | "[" [ КонстВыраж ] "]" ArraySize .

   Тип  =  УточнИдент .

Тип структура может быть объявлен как расширение другого типа структура. В примере 

struct T0 { number x; };

struct T1 : T0 { number y; }; 

T1 - (непосредственное) расширение T0, а T0 - (непосредственный) базовый тип T1 (см. Прил. A). Расширенный тип T1 состоит из полей своего базового типа T0 и полей, которые объявлены в T1. Все идентификаторы, объявленные в расширенной структуре, должны быть отличны от идентификаторов, объявленных в структуре(структурах) ее базового типа.  Нельзя экспортировать структуру, являющуяся потомком скрытой   (т.е. не экспортируемой) структуры.

Примеры объявлений типа структура: 

struct Point2* { number x, y; };

struct Point3 : Point2 { number z; };

7. Объявления переменных 

Объявления переменных дают описание переменных, определяя идентификатор и тип данных для них. 

ОбъявлениеПеременных = Тип СписокИдент “;”. 

СписокИдент  = идент ArraySize [СписокИниц] { "," идент ArraySize [СписокИниц] } .

СписокИниц =  [ "=" (Выражение | "{" СписокВыраж "}") ] . 

ArraySize = { "[" [ КонстВыраж ] "]" } .

Переменные типа структура имеют как статический тип (тип, с которым они объявлены - называемый просто их типом), так и динамический тип (тип их значения во время выполнения). Для параметров-переменных типа структура динамический тип может быть расширением их статического типа. Статический тип определяет, какие поля структуры доступны. Динамический тип используется для вызова связанных с типом процедур (см. 10.2). 

Примеры объявлений переменных (со ссылками на примеры из Гл. 6): 

number i, j, k, x, y;
 number a[100]; 

string s1, s2[100];

Point2  p2, x[30]; 

Point3  p3; 

8. Выражения 

Выражения - конструкции, задающие правила вычисления по значениям констант и текущим значениям переменных других значений путем применения операций и процедур-функций. Выражения состоят из операндов и операций. Круглые скобки могут использоваться для группировки операций и операндов. 

8.1 Операнды 

За исключением литералов (чисел или строк), операнды представлены обозначениями. Обозначение содержит идентификатор константы, переменной или функции. Этот идентификатор может сопровождаться селекторами, если обозначенный объект - элемент структуры. 

Обозначение  =  УточнИдент {"." идент | "[" Выражение "]" }. 

Если а - обозначение массива, a[e] означает элемент а, чей индекс - текущее значение выражения e. Тип e должен быть числовым типом.  Если r обозначает запись, то r.f означает поле f записи r или функцию f, связанную с динамическим типом r (Гл. 10.2). 

   Если обозначенный объект - константа или переменная, то обозначение ссылается на их текущее значение. Если он - функция, то обозначение ссылается на эту функцию, если только обозначение не сопровождается (возможно пустым) списком параметров. В последнем случае подразумевается активация функции и подстановка значения результата, полученного при ее исполнении. Фактические параметры должны соответствовать формальным параметрам как и при вызовах собственно функции (см. 10.1). 

Примеры обозначений (со ссылками на примеры из Гл. 7): 

      I                       number
      a[i]                   number
      x[3].y

      p3.y
      s1                    string
8.2 Операции 

В выражениях синтаксически различаются четыре класса операций с разными приоритетами (порядком выполнения). Операция ! имеет самый высокий приоритет, далее следуют операции типа умножения, операции типа сложения и отношения. Операции одного приоритета выполняются слева направо. Например, x-y-z означает (x- y) -z. 

Выражение = ИВыраж { "||" ИВыраж }.

 ИВыраж = ЕОтношение { "&&" ЕОтношение }.

 ЕОтношение = Отношение { ( "==" | "!=" ) Отношение } .

 Отношение = ПростоеВыражение { ( "<" | ">" | "<=" | ">=" ) ПростоеВыражение }.

 ПростоеВыражение =  Слагаемое { ("+" | "-") Слагаемое }.

 Слагаемое =  Множитель {( "*" | "/" | "%") Множитель }.

 Множитель =  ("!" | "+" | "-" ) Множитель | Число | string | "(" Выражение ")" | Обозначение ["(" [СписокВыраж] ")"] .

 Обозначение  =  идент {"." идент | "[" Выражение "]" }. 

 СписокВыраж  =  Выражение {"," Выражение}.

Предусмотренные операции перечислены в следующих таблицах. Некоторые операции применимы к операндам различных типов, обозначая разные действия. В этих случаях фактическая операция определяется типом операндов. Операнды должны быть совместимыми выражениями для данной операции (см. Прил. A). 

8.2.1 Логические операции 

	||   
	логическая дизъюнкция
	p || q      
	"если p, то TRUE, иначе q"

	&&
	логическая конъюнкция        
	p && q
	"если p то q, иначе FALSE"

	!
	Отрицание
	!p
	"не p"


Эти операции применимы к числовым операндам и дают результат типа number со значением 0 – если выражение ложно и не ноль иначе. 

8.2.2 Арифметические операции 

	+
	сумма

	-
	разность

	*
	Произведение

	/
	вещественное деление

	%   
	Остаток от деления


Операции +, -, *, и / применимы к операндам числовых типов.  Операция остаток от деления заключается в вычислении таких n и , что выполняется

x  % y = k.

 n* y+ k = x
n целое число, знак остатка совпадает со знаком x.

Операция + применима также к строкам и означает конкатенацию, т.е. присоединение всех символов второго операнда к концу строки – первого операнда. Конкатенация строки и числа означает, преобразование числа в строку и присоединение полученной строки к концу строки- первого операнда.

Примеры:

    5+ 3

12.5 % 3.5 = 2

-12.5 % 3 = -0.5

s1 = “? “

s1+ “так и надо” = “? так и надо”

s1+ 4  = “? 4”

s1+ 4+ 2  = “? 42”

8.2.4 Отношения 

	==
	Равно 

	!=
	не равно

	<
	Меньше

	<=    
	Меньшее или равно

	>
	Больше

	>=
	больше или равно


Отношения дают результат числового типа. Отношения ==, !=, <, <=, >, и >= применимы к числовым типам и  строкам. 

Примеры выражений (со ссылками на примеры из Гл. 7): 

	1991
	

	i / 3
	

	!p || q
	

	(i+j) * (i-j)
	

	i + x
	

	a[i+j] * a[i-j]
	

	(0<=i) && (I<100)
	

	t.key == 0
	

	w[i].name <= "John"    
	

	isstruct(t, Point3)
	

	
	


9. Операторы 

Операторы обозначают действия. Есть простые и структурные операторы. Простые операторы не содержат в себе никаких частей, которые являются самостоятельными операторами. Простые операторы - присваивание, вызов процедуры, операторы возврата, продолжения цикла и выхода. Структурные операторы состоят из частей, которые являются самостоятельными операторами. Они используются, чтобы выразить последовательное,  условное и повторное исполнение. Оператор также может быть пустым, в этом случае он не обозначает никакого действия. Пустой оператор добавлен, чтобы упростить правила пунктуации в последовательности операторов. 

Оператор  =  

    Обозначение ( "=" Выражение | "(" [СписокВыраж] ")" ) ";"

   | Тип идент VarList ";" 

   | ("if" |"если") "(" Выражение ")" Оператор  [ ("else" | "иначе") Оператор ]

   | ("while" | "цикл") "(" Выражение ")" Оператор

   | ("break" | "прервать")";"

   | ("continue" | "продолжить")";"

   | "{" Оператор { Оператор} "}" 

   | ("return" | "вернуть") Выражение ";"

   | ";"

   | ОператорSQL ";" .

9.1 Присваивания 

Присваивание заменяет текущее значение переменной новым значением, определяемым выражением. Выражение должно быть совместимо по присваиванию с переменной (см. Приложение. A). Знаком операции присваивания является "=", который читается "присвоить". 

Присваивание = Обозначение "=" Выражение. 

Если выражение e типа Te присваивается переменной v типа Tv, имеет место следующее: 

1. Если Tv и Te - структуры, то в присваивании участвуют только те поля Te, которые также имеются в Тv (проецирование); динамический тип v и статический тип v должны быть одинаковы, и не изменяются присваиванием; 

2. Если Тv это string[n], а e – string[m](строковая переменная длины m < n, v[i] присваиваются значения ei для i = 0 .. m-1, а v[m] получает значение 0X. 

Примеры присваиваний (со ссылками на примеры из Гл. 7): 

i = 0 
p = i == j 
x = i + 1 
a[i] = (x+y) * (x-y) 
t.key = i 
w[i+1].name = "John" 
t = c 

9.2 Вызовы функций 

Вызов функции активирует функцию. Он может содержать список фактических параметров, которые заменяют соответствующие формальные параметры, определенные в объявлении функции (см. Гл. 10). Соответствие устанавливается в порядке следования параметров в списках фактических и формальных параметров. Имеются два вида параметров: параметры-переменные и параметры-значения. 
     Если формальный параметр - параметр-переменная, соответствующий фактический параметр должен быть обозначением переменной. Если фактический параметр обозначает элемент структурной переменной, селекторы компонент вычисляются, когда происходит замена формальных параметров фактическими, то есть перед выполнением процедуры. Если формальный параметр - параметр-значение, соответствующий фактический параметр должен быть выражением. Это выражение вычисляется перед вызовом функции, а полученное в результате значение присваивается формальному параметру (см. также 10.1). 

ВызовФункции = Обозначение [ФактическиеПараметры]. 

Примеры: 

put(“%d “, i*2+1)       (* см. 10.1 *) 
 t.Insert("John")        (* см. 11 *) 

9.3 Составной оператор 

Последовательность операторов, разделенных точкой с запятой, означает поочередное выполнение действий, заданных составляющими операторами. 

ПоследовательностьОператоров = “{“ Оператор { Оператор} “}”. 

9.4 Операторы If 

   ОператорIf =   if (Выражение)  Оператор ; [else Оператор ; ] 


Операторы if задают условное выполнение входящих в них последовательностей операторов. Логическое выражение, предшествующие последовательности операторов, будем называть условием. Выражение вычисляется и если оно окажется не равным 0, то  выполняется связанный с этим условием оператор. Если условие не удовлетворено, выполняется оператор, записанный после слова else, если он имеется. 

Пример: 

D = b*b- 4*a*c;

if (d > 0) put(“два корня\n”);

else if (x == 0)  put(“один корень\n”);

else put(“нет корней\n”);

9.6 Операторы While 

Операторы while задают повторное выполнение оператора, пока выражение (условие) остается ненулевым. Условие проверяется перед каждым выполнением оператора. 

      ОператорWhile = “while” “(“ Выражение”)” Оператор . 


Пример: 

  // вычисление наибольшего общего делителя x, y
  x = 124078900;  y = 182045680;

  f = x; g = y;

  while (f != g) {

     if (f < g) g = g-f; else f = f-g;

  }

 // f = НОД(x, y)

9.7 Операторы Break

Оператор выхода из цикла break определяет завершение охватывающего оператора while и продолжение выполнения программы с оператора, следующего за оператором while. Оператор выхода связан с содержащим его оператором while контекстуально (т.е. с самым ближайшим  данного оператора оператором while), а не синтаксически.

      ОператорBreak = “break” “;” . 

Пример:

number a[100];

s = 0; i = 0;

while(1) {

   if (i == sizeof(a)) break;

   s = s+ a[i];

   i = i+1;

}

9.8 Операторы Continue 

     Оператор продолжения цикла обозначается словом continue. Он определяет переход к следующей итерации цикла while и продолжение выполнения программы с вычисления условия в операторе while. Оператор продолжения цикла выхода связан с содержащим его оператором while контекстуально (т.е. с самым ближайшим до данного оператора оператором while), а не синтаксически.

       ОператорContinue = “continue” “;” . 

Пример:

number a[100];

s = 0; i = 0;

while(1) {

   if (i == sizeof(a)) break;

   if (a[i] < 0) { i = i+ 1; continue; }

   s = s+ a[i];

   i = i+1;

}

9.10 Операторы возврата Return

Оператор возврата выполняет завершение функции или всей программы. Он обозначается словом return, за которым следует выражение. Тип выражения должен быть числовым. 
     Функции должны быть завершены оператором возврата, задающим значение результата. 

9.11 Операторы SQL
Оператор SQL предназначен для работы с базой данных. Оператор open создает курсор для выборки данных из запроса указанного в ПродолжениеSQL . Оператор fetch последовательно выбирает строки запроса одну за другой в поля, указанные Обозначение. Если курсор был открыт оператором open, то выборка осуществляется в строки, иначе если курсор был создан оператором opent, выборка осуществляется - для числовых полей в переменные числового типа, для полей типа даты в строки, для остальных типов полей в строки. Операторы INSERT, DELETE, UPDATE выполняют соответствующие операторы языка SQL. Оператор #sql в фигурных скобках содержит текст SQL запроса или программы на языке PL/SQL (версия Oracle) Transact SQL (версия SQL), которые могут быть связаны с переменными программы на языке Z2. Слова open, opent,close, fetch, delete, insert, update могут быть записаны в любом регистре (в соответствии с правилами SQL).

ОператорSQL = 

     ("open" | "opent") Обозначение ПродолжениеSQL

   | "close" Обозначение 

   | "fetch" Обозначение Обозначение {"," Обозначение }

   | "DELETE" ПродолжениеSQL

   | "INSERT" ПродолжениеSQL

   | "UPDATE" ПродолжениеSQL .

   | "#sql" [Обозначение ] “{“  SQLПредложение  “}”.
 ПродолжениеSQL = [СтрокаSQL] [SQLVAR] { СтрокаSQL [SQLVAR] [СтрокаSQL] } .

 SQLVAR = ":" ( идент | "{" Выражение "}" | "(" ( идент | "{" Выражение "}") ")") .

 SQLПредложение  = [СтрокаSQL] [":"идент] { СтрокаSQL [":"идент] [СтрокаSQL] } .

Оператор SQL может содержать подстановочные переменные SQLVAR. Механизм их обработки определяется директивой компилятора @B. Если эта директива не установлена, то обработка сводится к вычислению значения подстановычных переменных и замене их в ПродолжениеSQL на вычисленные значения. Если же директива @B установлена, то используется механизм Bind'ов.
Директива @B не установлена.

Подстановочная переменная может быть либо идентификатором, либо произвольным выражением, заключенным в фигурные скобки. Если тип идентификатора (выражения) числовой, то значение выражения, идентификатора во время выполнения подставляется вместо :SQLVAR, иначе если тип не числовой,  то значение подставляется в виде строки, окруженной апострофами. Если идентификатор или выражение указаны в круглых скобках, то строка подставляется без апострофов.

number t, k, out_n;

string out_str[100];

k = 10;  out_str = "_ВО_мес_премия";

open t select name from sl_pay where id = :k or sname = :out_str;

преобразовывается в

сформировать временную строку str_temp
str_temp := “select name from sl_pay where id = :k or sname = :out_str”;

заменить в str_temp :k на значение k (числовые величины преобразовываются в строку 15 знаков)

заменить в str_temp :out_str на значение out_str (будет подставлено 100 знаков на всю длину строки   out_str, недостающие знаки дополняются пробелами)

сгенерировать topen(str_temp, 0);

(Строка str_temp получится такой 

  “select name from sl_pay where id =      10 or sname = '_ВО_мес_премия '                                     ”)

 insert INTO sl_rep(id, s_id, fname) VALUES(sl_rep_seq.nextval, sessionid, :out_str);

преобразовывается в

сформировать временную строку str_temp
str_temp := “insert INTO sl_rep(id, s_id, fname) VALUES(sl_rep_seq.nextval, sessionid, :out_str)”;

заменить в str_temp :out_str на значение out_str (будет подставлено 100 знаков на всю длину строки   out_str, недостающие знаки дополняются пробелами)

сгенерировать DO_SQL(str_temp);

Так же, как и в случае с open числовые значения подставляются как строки длиною 15, а строки

Своей полной длиною.

Директива @B установлена.

Подстановочная переменная должна быть идентификатором переменной в программе.

Тип переменной должен быть number или string. 

open t select name from sl_pay where id = :k or sname = :out_str;

преобразовывается в
сгенерировать topen(“select name from sl_pay where id = :k or sname = :out_str”, 0);

сгенерировать tbind_num(t, “:k”, k);

сгенерировать tbind_str(t, “:out_str”, out_str);

insert INTO sl_rep(id, s_id, fname) VALUES(sl_rep_seq.nextval, sessionid, :out_str);

преобразовывается в
сгенерировать t1 = topen( “insert INTO sl_rep(id, s_id, fname) VALUES(sl_rep_seq.nextval, sessionid, :out_str)”, 3);

сгенерировать tbind_str(t1, “:out_str”, out_str);

сгенерировать texec(t1);

сгенерировать tclose(t1); 

Оператор #sql

#sql Обозначение {  SQLПредложение  } эквивалентно последовательности

   Обозначение = topen(SQLПредложение, 3);

   для каждой подстановочной переменной числового типа  tbind_num(Обозначение, “:ni”, ni);

   для каждой подстановочной переменной строкового типа tbind_str(Обозначение, “:si”, si);

#sql  {  SQLПредложение  } эквивалентно последовательности

   t1 = topen(SQLПредложение, 3);

   для каждой подстановочной переменной числового типа  tbind_num(t1, “:ni”, ni);

   для каждой подстановочной переменной строкового типа tbind_str(t1, “:si”, si);

   texec(t1);

   tclose(t1); 

Примеры:

 {

/*

CREATE OR REPLACE PROCEDURE

DROP_THIS_PROCEDURE(k number, out_n in out number, out_str out varchar2)

IS

BEGIN

  IF k = 1 THEN

     out_n := 43;  out_str := 'Ну и что';

  ELSE

     out_n := 37;  out_str := 'Да, было время';

  END IF;   

END;

/

*/

  #sql { begin DROP_THIS_PROCEDURE(:k, :out_n, :out_str); end; };

  put("out_n = %d out_str = %s", out_n, out_str);

  k = 10;  out_str = "_ВО_мес_премия";

  #sql t { select name from sl_pay where id = :k or sname = :out_str };

  fetch t, out_str;

  tclose(t);

  put("#sql k = %d out_str = %s", k, out_str);

  //@B+

  k = 10;  out_str = "_ВО_мес_премия";

  open t select name from sl_pay where id = :k or sname = :out_str;

  fetch t, out_str;

  tclose(t);

  put("open k = %d out_str = %s", k, out_str);

  delete from sl_rep where s_id = sessionid;

  insert INTO sl_rep(id, s_id, fname) VALUES(sl_rep_seq.nextval, sessionid, :out_str);

  k = 17;  out_str = "_ВО_мес_премия";

  //@B-

  opent t select fname from sl_rep where s_id = sessionid;

  fetch t, out_str;

  tclose(t);

  return 0;

}

  struct Point { number x, y; };

{

  string sname[20], name[50], s[160], date[40];

  var l, id, f, fi, fe;

  Point p;

  f = 300;  strcpy(date, "to_date(‘12.10.2002’, ‘DD.MM.YYYY’)");

  p.x = 14.41;

  opent tt

  SELECT id, sname, name

  FROM sl_pay

  WHERE pay_id > :f AND nvl(mdate, to_date('13.09.1994', 'DD.MM.YYYY')) > :(date)

  ORDER BY id;

  n = 0;

  while (1) {

     Fetch tt, l, sname, name;

     if (teot(tt)) break;

     …..

  }

  close tt;

  sname = “102030”;  name = “а это что такое”;

  INSERT INTO sl_rep(id, s_id, fname)

    VALUES (sl_rep_seq.nextval, :sname, :name);

  UPDATE sl_rep SET fname = 'I don''t know' WHERE id = 10;

  UPDATE sl_rep SET fname = :sname WHERE id = :{p.x} and n1 > 1;

  sput(s, "id, s_id, fname from sl_rep");

  open tt

  SELECT :(s);

     Fetch tt, s, sname, name;

  tclose(tt);

  return 0;

}

10. Объявления функций 

Объявление функции состоит из заголовка функции и  тела. Заголовок определяет имя функции и формальные параметры. Для связанных с типом функций в объявлении также определяется параметр-приемник (Тип идент). Тело содержит объявления и операторы.      Функции всегда возвращают результат. Если функция активизируется обозначением функции как часть выражения , то она возвращает результат, который является операндом этого выражения. Если же фунция активизируется как отдельный оператор, результат функции игнорируется Тело функции должно содержать оператор возврата, который определяет результат. 
     Все переменные, объявленные внутри тела функции, локальны в функции. Вызов функции изнутри ее объявления подразумевает рекурсивную активацию. 
     Объекты, объявленные в окружении процедуры, также видимы в тех частях процедуры, в которых они не перекрыты локально объявленным объектом с тем же самым именем. 

 ОбъявлФункции0 = "(" [ФормПарам {"," ФормПарам }] ")" . 

 ОбъявлФункции  = [Тип [идент]] "function" ИдентОпр 

                                   ОбъявлФункции0 (";" | "{" ПослОператоров "}"). 

 ОпережающееОбъявление = [Тип [идент]] "function" идент ОбъявлФункции0 ";".

Если объявление функции содержит параметр-приемник (Тип идент), функция рассматривается как связанная с типом (см. 10.2). Опережающее объявление служит чтобы разрешить ссылки на функцию, чье фактическое объявление появляется в тексте позже. Списки формальных параметров опережающего объявления и фактического объявления должны быть идентичны. 

10.1 Формальные параметры 

Формальные параметры - идентификаторы, объявленные в списке формальных параметров функции. Им соответствуют фактические параметры, которые задаются при вызове функции. Подстановка фактических параметров вместо формальных происходит при вызове функции. Имеются два вида параметров: параметры-значения и параметры-переменные, обозначаемые в списке формальных параметров отсутствием или наличием ключевого символа “&”. Параметры-значения это локальные переменные, которым в качестве начального присваивается значение соответствующего фактического параметра. Параметры-переменные соответствуют фактическим параметрам, которые являются переменными, и означают эти переменные. Параметры типа массив или структура всегда рассматриваются как параметры-переменные, Область действия формального параметра простирается от его объявления до конца блока функции, в котором он объявлен. Функция без параметров должна иметь пустой список параметров. Она должна вызываться обозначением функции, чей список фактических параметров также пуст. 

 ФормПарам  =  [Тип] ["&"] идент ArraySize. 

Пусть Tf - тип формального параметра f (не открытого массива) и Ta - тип соответствующего фактического параметра a. Для параметров-переменных Ta и Tf должны быть одинаковыми типами или Tf должен быть типом запись, а Ta - расширением Tf. Для параметров-значений а должен быть совместим по присваиванию с f (см. Прил. A). 
     Если Tf - открытый массив, то a должен быть совместимым массивом для f (см. Прил. A). Длина f становится равной длине a.

Примеры объявлений процедур: 

// сортировка массива методом выбора наименьшего

function asort(number a[], number n) {

  number i, j, k;

  number x;

  i = 0;

  while (i < n- 1) {

     k = i;  x = a[i];

     j = i+ 1;

     while (j < n) {

        if (a[j] < x) { k = j;  x = a[k]; }

        j = j+ 1;

     }

     a[k] = a[i];  a[i] = x;

     i = i+ 1;

  }

  return 0;

}
10.2 Функции, связанные с типом 

Функции могут быть ассоциированы с типом структура, объявленном в том же самом модуле. В этом случае говорится, что функции связаны с типом структура. Связь выражается типом приемника в заголовке объявления функции. Приемник должен быть параметром-переменной типа Т, где Т - тип структура. Функция, связанная с типом T, рассматривается как локальная для него. 

ОбъявлФункции  =  [Тип [идент]] "function" ИдентОпр

                                  ОбъявлФункции0 (";" | "{" ПослОператоров "}"). 

Если функция P связана с типом T0, она неявно также связана с любым типом T1, который является расширением T0. Однако функция P' (с тем же самым именем что и P) может быть явно связана с T1, перекрывая в этом случае связывание c P. P' рассматривается как переопределение P для T1. Формальные параметры P и P' должны совпадать (см. Прил. A).  Если P и T1 экспортируются, то P' тоже должна экспортироваться. Если P не экспортируется, то P' тоже не должна экспортироваться. Если v - обозначение, а P - связанная функция, то v.P обозначает функцию P, связанную с динамическим типом v. Заметим, что это может быть функция, отличная от той, что связана со статическим типом v. v передается приемнику функции P согласно правилам передачи параметров, определенным в Главе 10.1. 
В опережающем объявлении связанной функции и в фактическом объявлении функции параметр-приемник должен иметь одинаковый тип. Списки формальных параметров в обоих объявлениях должны быть идентичны. 

Примеры:

struct FxIntegral { number s; };

FxIntegral function f(number x1) { return x1; }

FxIntegral x function Integral(a, b, s1) {

  number r, s;

  r = 0;  s = s1;

  if (a >= b || s1 <= 0) return 0;

  x1 = a;  x2 = a+ s1;  if (x2 > b) { x2 = b; s = b- x1; }

  f1 = x.f(x1);   f2 = x.f(x2);

  while (x1 < b) {

     r = r+ (f1+ f2)/2 * s;

     x1 = x2; x2 = x2+ s1;  if (x2 > b) { x2 = b; s = b- x1; }

     f1 = f2; f2 = x.f(x2);

  }

  return r;

}

struct ConstIntegral: FxIntegral { };

ConstIntegral x function f(number x1) { return 5; }

struct LinIntegral: FxIntegral { };

LinIntegral x function f(number x1) { return 3*x1+ 5; }

{

 ConstIntegral f1;

 x1 = 0; x2 = 3;

 r = f1.Integral(x1, x2, 0.01);

 put("Интеграл функции х = 5 на участке %d..%d равен %d\n", x1, x2, r);

 LinIntegral f2;

 r = f2.Integral(x1, x2, 0.01);

 put("Интеграл функции 3*х+5 на участке %d..%d равен %d\n", x1, x2, r);

 вернуть 0;

}

10.3 Стандартные функции
Следующая таблица содержит список стандартных функций. Некоторые функции - обобщенные, то есть они применимы к операндам нескольких типов. Буква v обозначает переменную, x и n - выражения, T - тип. Для стандартных функций массив типа number и двумерный массив типа number могут подставляться везде, где требуется строка.

sizeof (v)   размер переменной или выражения v в 8-байтных словах

sizeof v  то же что и sizeof (v)

isstruct(v, T)  возвращает 1 если  "динамический тип v есть T (или расширение T)" и называется проверкой типа. Проверка типа применима, если  v - параметр-переменная типа структура и если T - расширение статического типа v.

strlen(string s[])  Возвращает длину строки без нулевых символов

strtrunc(string s[], number len) -- установить 0 в позицию len строки s

strdel(string s[], pos, n) - удалить n символов строки s начиная с позиции pos
strins(string dst[], pos, string src) - вставить строку src в строку dst, начиная

    с позиции pos

strcpy(string s1[], string s2[])  Скопировать строку s2 в s1.

          Предполагается, что размер массива s1 достаточен для размещения строки s2.

strncpy(string s1[], string s2[], number n)
     Скопировать в s1 не более n символов из s2.

strcat(string s1[], string s2[])  Добавить строку s2 к концу строки s1.

strncat(string s1[], string s2[], number n)  Добавить к s1 не более n символов из s2.

N = strcmp(string s1[], string s2[])  Посимвольное сравнение строк s1 и s2

     Если s1 < s2, то n = -1;

             s1 = s2, то n = 0;

             s1 > s2, то n = 1.

     Нулевой символ считается меньше любого знака

      strcmp("ну и", "ну и что") = -1

N = strncmp(string s1[], string s2[], number nn)  Посимвольное сравнение строк s1 и s2, не более 

         чем по  nn символам

ret = strstr(string s1[], string s2[])  Возвращает позицию символа, начиная с которого

         строка s2 целиком входит в s1, или –1 (если не найдена).

ret = strnstr(string s1[], number pos, string s2[])  Возвращает позицию символа, начиная с которого

         строка s2 целиком входит в строку s1с позиции pos, или -1 (если не найдена).

strnum(string s[], number &n)  Преобразовать строку s в число n, если это возможно

numstr(number n, string s[])  Преобразовать число n в строку s.

strupr(string s[])  Привести строку s к верхнему регистру, т.е. преобразовать все маленькие 

       буквы в большие.

strlow(string s[])  Привести строку s к нижнему регистру, т.е. преобразовать все большие 

       буквы в маленькие
strscpy(string s1[], string s2[], number n)  Скопировать в s1 конец строки s2 с позиции n.

strdcpy(string s1[], string s2[], number n)  Заменить в строке s1 символы с позиции n на 

                      символы из строки s2.
ltrim(string s1[])   Отсечение пробелов от строки s1 слева

rtrim(string s1[])  Отсечение пробелов от строки s1 справа
strset(string s1[], string s2[], number n) Установить строку s1 как n повторений строки s2

                           strset(s1, “Qu”, 3);

strn2cpy(string s1[], string s2[], number pos, number n) 

        Скопировать в строку s1 подстроку из s2 с позиции pos длиной n

        strn2cpy(s1, “KeyBoard”, 3, 2);  //  s1 станет равной “Bo”
n = ord(s1)  s1 предполагается кодом символа и функция возвращает код ASCII

                    данного символа   ord(“Б”) = 193  // для Windows- клиента 

chr(n, s1)    Обратная функция для ord

                     chr(193, s1) ;   // s1 == “Б”

t=topen(string SQL-предложение[],number f)

Открывает виртуальную таблицу, соответствующую select-предложению языка SQL из SQL-предложение. Если второй параметр (f)  равен 0, то результаты чтения функцией tget записываются в строки, иначе если f = 1 или 3, то для числовых полей результаты записываются в числовые переменные, а остальные в строки

Возвращаемое значение есть некоторый индекс, используемый в других функциях (tget, tclose, texec, tbind_num,…).

Если t < 0, то таблица не открыта.

n=tget(t,стр1,стр2,...)

   Читает следующую строку из таблицы t и значения полей записывает в переменные стр1, стр2,...

    Первое поле в стр1, второе в стр2 и т.д.

   Функция имеет переменное число параметров (количество параметров должно быть меньше

    или равно количеству полей, зачитанных при открытии таблицы).

    Если курсор был открыт с типом 0, то тип переменных стр1, стр2,... должен быть строковым,

    иначе тип переменной должен соответствовать типу соответствующего поля SQL запроса 

    (числовой - числовым, строковые остальным типам полей).

    Возвращает 1, если в таблице есть данные, 0 если таблица пуста или все данные уже прочитаны

    и -1 при ошибке.

    t = topen("select Auto_Card, Num_Tab, Name from Card",0);

    while(1) {

        var card[2], ntab[1];

        if (0 >= tget(t,card,ntab)) break;

        put(“ИдФизЛица  %s,  Таб.N  %s \n”, card, ntab);

    }

    tclose(t);

teot(t)    Возвращает 0 если еще есть записи в таблице и 1 иначе

tclose(t)  Закрывает таблицу t. Должна быть парной topen и обязательно использоваться,

                т.к. при открытии таблицы динамически выделяются некоторые ресурсы.

DO_SQL(string строка[])  

       Выполняет SQL-предложение строка

       DO_SQL("delete from Card where Auto_Card=10");

texec(number t) 

      Выполнить SQL-предложение, заданное при выполнении topen.

tbind_num(number t, string par[], number &val)

   Большинство SQL операторов, содержит данные которые надо передавать серверу. Например,   следующий оператор, добавляющий работника, содержит в операторе значения кода работника и его фамилию

   INSERT INTO emp VALUES(2310, ‘Петров В.Н.’)

   Чтобы добавить Сидорова надо выполнить предложение

   INSERT INTO emp VALUES(2311, ‘Сидоров А.Н.’)

   SQL оператор при этом должен перекомпилиться на сервере. Этот процесс может отнимать    много времени. Чтобы обойти это, в SQL оператор мы подставим переменные связи. 

  INSERT INTO emp VALUES(:kod, :family)

  В языке Z2 нам необходимо каждой такой связной переменной сопоставить переменную языка,     причем числовой – числовую, символьной- строку.

  Этот   процесс   называется   СВЯЗЫВАНИЕМ   ПЕРЕМЕННЫХ   (BIND).

  Т.к.мы указываем адрес  (связывание по  ссылке), нам  не требуется повторять этап связывания  перед повторным  исполнением предложения.  Вы  можете  просто  изменять  значение  по  этому   адресу, а ORACLE (SQL) находит это значение перед каждым  исполнением, используя адрес памяти, предоставленный перед связыванием.

tbind_num(number t, string par[], number &val) 
для курсора t связывает SQL переменную par с числовой переменной языка val
tbind_str(number t, string par[], string val[])

для курсора t связывает SQL переменную par с строковой переменной языка val
string fname[80];

number t, i2, i;

{

  t = topen("insert into sl_rep(id, s_id, i2a, fname, i2b) "

                  " values(sl_rep_seq.nextval, sessionid, :I2, :fn, :i)", 1);

  tbind_num(t, ":I2", i2);

  tbind_str(t, ":fn", fname);

  tbind_num(t, ":i", i);

  i2 = 1;  fname = "Привет";  i = 4;

  texec(t);

  i2 = 2;  fname = "Пока";  i = 4;

  texec(t);

  i2 = 3;  fname = "Такие вот дела";  i = 2;

  texec(t);

  tclose(t);

  t = topen("select i2a, fname from sl_rep "

                 "where s_id = sessionid and i2b = :i order by i2a", 1);

  tbind_num(t, ":i", i);  i = 4;

  while (1) {

     if (tget(t, i2, fname) <= 0) break;

     put("i2a = %d  fname = %s", i2, fname);

  }

  tclose(t);

  return 0;

}

11. Модули 

Модуль - совокупность объявлений констант, типов, переменных и функций вместе с последовательностью операторов, предназначенных для присваивания начальных значений переменным. Модуль представляет собой текст, который является единицей компиляции. 

В системе БОСС-Кадровик модуль является Z процедурой. Именем модуля является имя Z процедуры.

Z2 = [СписокИмпорта] ПослОбъявл "{" ПослОператоров "}" .

 СписокИмпорта = "import" идент ["=" идент ]{"," идент ["=" идент ] }.

Список импорта определяет имена импортируемых модулей. Если модуль A импортируется модулем M, и A экспортирует идентификатор x, то x упоминается внутри M как A.x. Если A импортируется как B=A, объект x должен вызываться как B.x. Это позволяет использовать короткие имена-псевдонимы в уточненных идентификаторах. Модуль не должен импортировать себя. Идентификаторы, которые экспортируются (то есть должны быть видимы в модулях-импортерах) нужно отметить экспортной меткой в их объявлении (см. Главу 4). 
Последовательность операторов после символа “{“ выполняется, когда модуль добавляется к системе (загружается). Это происходит после загрузки импортируемых модулей. Отсюда следует, что циклический импорт модулей запрещен. 

Примеры:

//Модуль для работы со строками Strings
// вернуть длину строки s (число символов до завершающего нуля)

function Length*(string s[]) {

  return strlen(s);

}

// записать 0 в позицию len строки s

function SetLength*(string s[], number len) {

  return strtrunc(s, len);

}

// Вставить строку source в строку dst, начиная с позиции pos.

// 0 <=pos<=Length(dst)).  Если pos = Length(dst), source добавляется к концу dst.

function Ins*(string dst[], number pos, string source[]) {

  return strins(dst, pos, source);

}

// добавить строку extra к концу строки dest. То же, что и Ins(dst, s, Length(dst)).

function Append*(string dest[], string extra[]) {

  strcat(dest, extra);

  return 0;

}

// Удалить из строки s n символов, начиная с позиции pos 

// pos (0 <= pos <= Length(s)).

// Если n > Length(s) - pos, то новая длина строки s есть pos.

function Del*(string s[], number pos, number n) {

  return strdel(s, pos, n);

}

// В строке dest, начиная с позиции pos, заменить все симолы на символы строки source

// Равносильно Delete(dest, pos, Length(source)); Insert(dest, pos, source).

function Replace*(string dest[], number pos, string source[]) {

  strdcpy(dest, source, pos);

  return 0;

}

// Записать в строку dest подстроку длиною n строки source, начиная с позиции pos

// 0 <= pos <= Length(source). Если n > Length(source) - pos, в dest записывается
// часть строки source, начиная с позиции pos и до конца строки, т.е. Length(src) -1.

function Extract*(string dest[], string source[], number pos, number n) {

  return strn2cpy(dest, source, pos, n);

}

// Возвращает позицию первого вхождения строки pattern в строке s, начиная с позиции pos

// Если строка не найдена вернуть -1

function Pos*(string s[], number pos, string pattern[]) {

  return strnstr(s, pos, pattern);

}

// Преобразовать все символы строки в прописные

function Cap*(string s[]) {

  strupr(s);

  return 0;

}

{

  return 0;

}

Использование модуля Strings

import str=Strings;

{

  string s[100], d;

  s = "12345670qwerty0asdfgh0zxcvbn";

  d = "12345670qwerty0asdfgh0zxcvbn";

  put("Length('%s') = %d\n", s, str.Length(s));

  str.SetLength(s, 23);

  put("Length('%s') = %d\n", s, str.Length(s));

  strcpy(s, "+-*/-+"); put("\ns = '%s'\n", s);

  pos = 10; str.Ins(d, pos, s);  put("Ins(d, %d, s) d = '%s'\n", pos, d);

  str.Append(d, s);   put("Append(d, s)  d = '%s'\n", d);

  n = 12; str.Del(d, pos, n);   put("Del(d, %d, %d) d = '%s'\n", pos, n, d);

  str.Replace(d, pos, "908070");  put("Replace(d, %d, '908070')  d = '%s'\n", pos, d);

  strcpy(s, "12345670qwerty0asdfgh0zxcvbn"); put("\ns = '%s'\n", s);

  str.Extract(d, s, pos, n);  put("Extract(d, s, %d, %d) d = '%s'\n", pos, n, d);

  put("s = '%s' Pos(s, %d, 'y0asd') = %d\n", s, pos, str.Pos(s, pos, "y0asd"));

  str.Cap(s);  put("s = '%s'\n", s);

 return 0;

}

Приложение A: Определение терминов 

Одинаковые типы 

Две переменные a и b с типами Ta и Tb имеют одинаковый тип, если 

1. Ta и Tb оба обозначены одним и тем же идентификатором типа, или 

2. a и b появляются в одном и том же списке идентификаторов переменных, полей записи или объявлении формальных параметров и не являются открытыми массивами. 

Равные типы 

Два типа Ta, и Tb равны, если 

1. Ta и Tb - одинаковые типы, или 

2. Ta и Tb - типы открытый массив с равными типами элементов

Расширение типов (базовый тип) 

В объявлении типа struct Tb: Ta{ ... }, Tb - непосредственное расширение Ta, а Ta - непосредственный базовый тип Tb. Тип Tb есть расширение типа Ta (Ta есть базовый тип Tb) если 

1. Ta и Tb - одинаковые типы, или 

2. Tb - непосредственное расширение типа, являющегося расширением Ta 

Совместимость по присваиванию 

Выражение e типа Te совместимо по присваиванию с переменной v типа Tv, если выполнено одно из следующих условий: 

1. Te и Tv - одинаковые типы; 

2. Te и Tv - типы запись, Te есть расширение Tv, а v имеет динамический тип Tv; 

3. Tv – string[n], e – string[m]. Контроль размерности происходит во время выполнения; 

Совместимость массивов 

Фактический параметр a типа Ta является совместимым массивом для формального параметра f типа Tf если 

1. Tf и Ta - одинаковые типы или 

2. Tf - открытый массив, Ta - любой массив, а типы их элементов - совместимые массивы или 

3. f - параметр-значение типа string[n], а фактический параметр a - строка. 

Совместимость выражений 

Для данной операции операнды являются совместимыми выражениями, если их типы соответствуют следующей таблице (в который указан также тип результата выражения). Символьные массивы, которые сравниваются, должны содержать в качестве ограничителя 0X. Тип T1 должен быть расширением типа T0: 

	операция 
	первый операнд
	второй операнд
	тип результата

	+ - *  / %
	числовой
	числовой
	числовой 

	/ 
	числовой
	числовой
	 числовой 

	+ 
	 строка
	 строка
	 Строка

	|| &&  ! 
	числовой 
	числовой
	числовой

	== != < <= > >=   
	числовой
 строка
	числовой
 строка
	числовой
строка


Совпадение списков формальных параметров 

Два списка формальных параметров совпадают если 

1. они имеют одинаковое количество параметров, и 

2. параметры в соответствующих позициях имеют равные типы, и 

3. параметры в соответствующих позициях - оба или параметры-значения или параметры-переменные. 

Приложение B: Синтаксис Z2 

Z2 = [СписокИмпорта] ПослОбъявл "{" ПослОператоров "}" .

 СписокИмпорта = "import" идент {"," идент }.

 ПослОбъявл  =  { 

  "struct" ИдентОпр [ ":" Тип ] "{" { Тип ИдентОпр ArraySize { "," ИдентОпр ArraySize } ";" } "}" ";"

   | Тип [идент] ( VarList ";" | ОбъявлФункции )

   | ОбъявлФункции

   | "const" ИдентОпр "=" КонстВыраж { "," ИдентОпр "=" КонстВыраж } ";"

   } . 

ArraySize = { "[" [ КонстВыраж ] "]" } .

VarList   = ArraySize [СписокИниц] { "," идент ArraySize [СписокИниц] } .

 СписокИниц =  [ "=" (Выражение | "{" СписокВыраж "}") ] . 

 ОбъявлФункции0 =  "(" [ФормПарам {"," ФормПарам }] ")" . 

 ОбъявлФункции  =  "function" ИдентОпр ОбъявлФункции0 (";" | "{" ПослОператоров "}"). 

 ФормПарам  =  [Тип] ["&"] идент ArraySize.

 Тип  =  УточнИдент .

 ПослОператоров  =  Оператор { Оператор}. 

 Оператор  =  Обозначение ( "=" Выражение | "(" [СписокВыраж] ")" ) ";"

   | Тип идент VarList ";" 

   | ("if" |"если") "(" Выражение ")" Оператор  [ ("else" | "иначе") Оператор ]

   | ("while" | "цикл") "(" Выражение ")" Оператор

   | ("break" | "прервать")";"

   | ("continue" | "продолжить")";"

   | "{" Оператор { Оператор} "}" 

   | ("return" | "вернуть") Выражение ";"

   | ";"

   | ОператорSQL ";" .

 ОператорSQL = 

     "open" Обозначение ПродолжениеSQL
   | "close" Обозначение 

   | "fetch" Обозначение Обозначение {"," Обозначение }

   | "DELETE" ПродолжениеSQL
   | "INSERT" ПродолжениеSQL
   | "UPDATE" ПродолжениеSQL .

   | "#sql" [Обозначение ] “{“  SQLПредложение  “}”.
 ПродолжениеSQL = [СтрокаSQL] [SQLVAR] { СтрокаSQL [SQLVAR] [СтрокаSQL] } .

 SQLVAR = ":" ( идент | "{" Выражение "}" | "(" ( идент | "{" Выражение "}") ")") .

SQLПредложение  = [СтрокаSQL] [":"идент] { СтрокаSQL [":"идент] [СтрокаSQL] } .

 КонстВыраж  =  Выражение. 

 Выражение = ИВыраж { "||" ИВыраж }.

 ИВыраж = ЕОтношение { "&&" ЕОтношение }.

 ЕОтношение = Отношение { ( "==" | "!=" ) Отношение } .

 Отношение = ПростоеВыражение { ( "<" | ">" | "<=" | ">=" ) ПростоеВыражение }.

 ПростоеВыражение =  Слагаемое { ("+" | "-") Слагаемое }.

 Слагаемое =  Множитель {( "*" | "/" | "%") Множитель }.

 Множитель =  ("!" | "+" | "-" ) Множитель | Число | Строка | "(" Выражение ")" | Обозначение ["(" [СписокВыраж] ")"] .

Обозначение  =  УточнИдент {"." идент | "[" Выражение "]" }.

 СписокВыраж  =  Выражение {"," Выражение}.

 Число = number.

 УточнИдент =  [идент "."] идент. 

 ИдентОпр = идент [ "*" ].
Приложение C: Особенности реализации Z2 в БОСС-Кадровике

В язык добавлены внешние константы и реализовано большое количество специфических(имеющих отношение в основном к БОСС-Кадровику) встроенных функций.

Внешние Константы

Внешние константы – это поля sname - таблиц входимости, видов оплат, режимов работы и систем оплаты. Например, _тВх_премия или _ВО_оклад. Их значения соответственно равны номеру таблицы, коду вида оплаты, коду режима работы и коду системы оплаты.

Иначе говоря, зачитывается в версии Oracle 
     select sname, id, 'sl_Pay_tbl' from sl_Pay_tbl 

     union 

     select sname, id, 'sl_Pay' from sl_Pay 

     union 

     select sname, id, 'hr_SysPay' from hr_SysPay 

     union 

     select sname, id, 'hr_Regim' from hr_Regim

     ORDER BY 1     

     ;

или в версии SQL
   select sname, Code_Pay_tbl, 2 from typ_Pay_tbl(NOLOCK)

   union all

   select sname, Code_Pay, 1 from typ_Pay(NOLOCK)

   union all

   select sname, Code_Regim, 4 from typ_Regim(NOLOCK)

и эти идентификаторы (значения полей sname)  считаются объявленными

Рекомендуется краткие имена 

   видов оплаты начинать с _ВО_

   таблиц входимости  с _тВх_

   систем оплаты  с _ст_

   режимов работы с _рр_  

К именам внешних констант предъявляются те же требования, что и к именам переменных, однако их длина ограничивается 15 знаками.

Общецелевые функции

n = Z_платформа() 

	Описание
	возвращает числовую константу  Z_MsSQL для Ms SQL,   Z_Oracle для Oracle


	Возвращаемое значение
	числовая константа  Z_MsSQL для Ms SQL,   Z_Oracle для Oracle

	Пример
	   if (Z_платформа() == Z_MsSQL)

      sput(st, 

         "select count(Auto_card) from people “

         "where id_firm = %d ", idFirm);

   else if (Z_платформа() == Z_Oracle)

      sput(st, 

         "select count(card_id) from pr_emp “

         "where firm_id = %d ", idFirm);




n = load_dbf(file, tname, mw, mv, ask)

только в версии для Oracle
	Описание
	Загрузить DBF-файл file в таблицу tname
имя таблицы будет преобразовано в большие буквы

mw  -  1- кодировка WIN1251, 0- Dos866

mv  -   1- строки VARCHAR2, 0- CHAR

ask -   1- задать вопрос для удаления таблицы tname (если она уже есть),

           0 – таблица удалится без вопроса



	Возвращаемое значение
	Возвращает 0 при успехе и не 0 иначе



	Пример
	load_dbf("a10.dbf", “hr_imp_a10”, 0, 1, 0);


n = to_dbf(file, kodirovka, SQL_stmt)

	Описание
	По select-предложению SQL_stmt сформировать DBF-файл file в кодировке 866, если kodirovka = 0, иначе в кодировке Windows.  Формат файла DBASE III

С марта 2002г. kodirovka рассматривается как десятичное число abc
      c - 1- кодировка 1251, 0 - кодировка 866

     b  - 1- целые беззнаковые, 0 - знаковые

    a   - 1- нецелые числа беззнаковые, 0 - знаковые

	Возвращаемое значение
	Возвращает 0 при успехе и не 0 иначе



	Пример
	to_dbf("a10.dbf", 0, "select * from st_categ");


n = to_Excel(string file, string list, string SQL_stmt) 
	Описание
	Только для версии на платформе Oracle
По select-предложению SQL_stmt сформировать Excel-файл file на сервере приложений в каталоге пользователя. Если list не пусто, то результат запишется в соотвествующий лист данного файла. Поля mdate , если они типа DATE, выводятся с маской dd.mm.yyyy hh24:mi:ss, остальные поля типа DATE выводятся без времени. Числовые поля ID или заканчивающиеся на _ID, выводятся как целые шириною 10 символов, остальные – как числа с 4 знаками после запятой

	Возвращаемое значение
	Возвращает 0 при успехе и не 0 иначе



	Пример
	to_Excel("a10.xls", “”, "select * from st_categ");

to_Excel("a10.xls", “regim”, "select * from hr_regim");




Round(x,r)

	Описание
	Округлить значение x до величины r.

	Пример
	round(21502,1000) = 22000

round(21300,1000) = 21000


Trunc(x,r)

	Описание
	Отбросить незначащий разряд x до величины r.

	Пример
	trune(21502,1000) = 21000

trune(21300,1000) = 21000


Функция min(x,y)

	Описание
	Возвращает минимум из x и y.


Функция max(x,y)

	Описание
	Возвращает максимум из x и y.


beep()

	Описание
	Выдать звуковой сигнал


host(команда_ОС)

	Описание
	Передать на выполнение операционной системе строку команда_ОС
host(“notepad.exe rep.txt”);  // вызвать редактор notepad для файла         rep.txt


client_host(str [,number блокировать])

	Описание
	Передать на выполнение операционной системе компьютера текущего пользователя строку команда_ОС. Если блокировать  не ноль, то приостановиться до заверешения выполнения команды. Умолчание блокировать  = 1.
client_host(“notepad.exe rep.txt”);  // вызвать редактор notepad для файла         rep.txt



abs(x)

	Описание
	Возвращает абсолютное значение x


SetArray(МАССИВ, значение1, значение2, …, значениеN)

	Описание
	Устанавливает значение I-го элемента массива МАССИВ в значение значениеI-1

Равносильно
I = 0;

while (I < n) {

   МАССИВ[I] = значениеI; I = I+ 1;

}

	Пример
	Var a[3];

SetArray (a, 3, 5, 7);


sput(s, формат, пер1, пер2, …

	Описание
	Сформировать строку s из переменных пер1, пер2, … согласно строке формата формат. Строка формата представляет собой комбинацию обычных символов и спецзнаков %,за которыми следует символ формата Допустимые символы формата d и s

%d - означает ожидание числа на вводе

%s - означает ожидание строки на вводе

strcpy(s2, “Февраль”);

sput(s, “Месяц %s  Год %d”, s2, 1995);

Будет сформирована строка s “Месяц Февраль Год 1995”

Более обще формат для строки имеет вид %[знак][длина]s Квадратные скобки означают что указанного поля может не быть

%10s  первые 10 символов строки. Если строка короче будет дополнена

пробелами справа до длины 10

%+10s  Строка будет выровнена вправо 

Для чисел

%[знак][длина][.точность]d

Знак - означает выравнивание влево, + или пусто - выравнивание вправо

	Пример
	Пусть n = 1234.567;

%d            bb1235.567

%10d          bbbbbb1235

%10.2d        bbb1234.58

%-10.2d       1234.58bbb   выравнивание влево

%.2d          1234.58

          b это знак пробела


z_debug(формат,пер1,...)

	Описание
	Функция аналогична sput

Однако работает только в состоянии отладка и записывает свои результаты в протокол расчета(устанавливается в дополнительной настройке расчета)

 В обычном режиме игнорируется


is_z_debug()

	Описание
	Возвращает 
1 – если программа работает только в состоянии отладка и записывает свои

      результаты в протокол расчета

0 – если протокол расчета не включен


z_trace(формат,пер1,...)

	Описание
	Функция аналогична z_debug
Однако работает только в состоянии «включена трассировка».

 (устанавливается в дополнительной настройке расчета)

 Результаты функции записываются в таблицу sl_trace и отображаются в формах просмотра расчета

В обычном режиме игнорируется.

Результаты z_debug как правило предназначаются разработчику, а z_trace – конечному пользователю.

В момент расчета заработной платы результат привязывается к текущей расчетной строке, в остальных случаях функция работает также как z_debug(…)


z_trace_pay(ИдРасчетнойСтроки, формат,пер1,…)

	Описание
	Функция аналогична z_trace(формат,пер1,...)

Но результат привязывается не к текущей строке расчета и соответственно виду оплаты, а к расчетной строке с номером ИдРасчетнойСтроки.


Функции для работы с датами

Предполагается что даты - это массив из трех элементов (день, месяц, год). Период между датами также может представляться в виде массива (количество дней, месяцев, лет).

В дальнейшем описании предполагается описание s

var d1[3], d2[3], pd[3], n;

Дата(день,месяц,год,d1)

	Описание
	Установить дату d1 с соответствующими значениями:

Дата(1,4,1994,d1);

Дата(3,3,1997,d2);


Период(дней,месяцев,лет,pd)

	Описание
	Аналогично функции 1, только установить период.


ДатаМинусДата(d2,d1)

	Описание
	Возвращается количество дней между дата d1 и d2.

ДатаМинусДата(d2,d1) = 1097 дней.


ДатаПлюсДни(d1,n,d2)

	Описание
	d2 устанавливается в дату, которая наступит через n дней после даты d1.

Если n = 10, то

ДатаПлюсДни(d1,10,d2) установит d2 в 1.4.1994.


ДатаМинусДни(d1,n,d2)

	Описание
	d2 устанавливается в дату, которая меньше d1 на n дней.

ДатаМинусДни(d1,1,d2) установит d2 в 22.3.1994.


ПериодМеждуДатами(d1,d2,pd

	Описание
	pd устанавливается равным периоду между датами d1, d2, выраженному в днях месяцах и годах.

ПериодМеждуДатами(d1,d2,pd) равен 2 для 11 месяцев и 2 лет, т.е. 

pd[0]=2; pd[1]=11; pd[2]=2;


ДатаПлюсПериод(d1,pd,d2)

	Описание
	d2 устанавливается в дату, равную d1 плюс период pd.

Если pd - (2,11,2), d1 (1,4,1994), то

ДатаПлюсПериод(d1,pd,d2) установит d2 в (3,3,1997).


ДатаМинусПериод(d1,pd,d2)

	Описание
	d2 устанавливается в дату, равную d1 минус период pd.

Если pd - (6,3,2), d1 (3,3,1997), то

ДатаМинусПериод(d1,pd,d2) установит d2 в (25,11,1994).


ДатаРавнаДате(d1,d2)

	Описание
	Возвращает 1, если дата d1 равна дате d2, иначе - 0.


ДатаМеньшеДаты(d1,d2)

	Описание
	Возвращает 1, если дата d1 меньше даты d2, иначе - 0.


Дополнительные функции для работы с БД

Функцией topen можно открыть курсор не только для SELECT предложения, но и для предложений INSERT, UPDATE, DELETE. SQL предложение также может содержать переменные связи (bind переменные). В версии для SQL это возможно только при втором параметре равном 3, в версии для Oracle при всех значениях параметра открытия курсора.

В версии для SQL сервера SQL – предложение должно быть правильным предложением на Transact SQL с возможностью использования расширений X-языка (при втором параметре не равном 3). При втором параметре равном 3 допустимы лишь следующие переменные X языка: CURSTATION.

В версии для Oracle SQL – предложение может также быть PL/SQL блоком.

tfirst(t)

	Описание
	Переход к началу таблицы t.


n=tcount(t)

	Описание
	Возвращает количество записей в таблице t.


n=tcoln(t)

	Описание
	Возвращает количество столбцов в запросе в таблице t.


n=tcolt(t, n)

	Описание
	Возвращает тип колонки n в запросе в таблице t.

1 – числовое поле

2 – строковое поле

3 – поле даты, времени (DATE, DATETIME)

4 – поля LOB (CLOB,BLOB,LONG, LANG RAW- Oracle, TEXT, IMAGE – SQL server)


n=tcolsz(t, n)

	Описание
	Возвращает отображаемый размер колонки n в запросе в таблице t.


n=tcolv_null(t, n)

	Описание
	Возвращает признак заполненности значения поля колонки n в запросе в таблице t после чтения строки (tget или fetch).

1 – поле пусто (NULL)
0 – поле не пусто


n= tbind_num_array(number t, string par[], number nval[])  для версии Oracle
	Описание
	для курсора t связывает SQL переменную par с числовым массивом  nval
Возвращает 0 – при успехе и не 0 при ошибке


n= tbind_str_array(number t, string par[], string sval[])             для версии Oracle
	Описание
	для курсора t связывает SQL переменную par с строковым массивом sval
Возвращает 0 – при успехе и не 0 при ошибке

PL/SQL
CREATE OR REPLACE PROCEDURE

get_trans(emp number, out_n in out number, wage out DBMS_SQL.Number_Table, period out DBMS_SQL.Varchar2s)

Z2
function aa(emp, n, wage[], string period[][48]) {

  t = topen("begin get_trans(:emp, :n, :wage, :period); end;", 1);

  tbind_num(t, ":emp", emp);

  tbind_num(t, ":n", n);

  tbind_num_array(t, ":wage", wage);

  tbind_str_array(t, ":period", period);

  texec(t);

  close t;

  i = 0;

  while (i < n) {

    dput(">t_arr.t", "...i = %d  wage = %d  period = %s\n", i, wage[i], period[i]);

    i = i+ 1;

  }

  return 0;

}




getLong(s, s2, len)

	Описание
	Прочитать первые len позиций MEMO поля s  в массив s2. Поле s должно быть в формате ИмяТаблицы/ИмяПоляMEMO/ИмяПоляИД/ЗначениеЭтогоПоля

	Пример
	var prog[3000], s[14], s_ins[14];

sput(s, "sl_pay/prog/id/%d", 12);

getLong(s, prog, 1800);

put("Программа вида оплаты с кодом 12\n%s", prog);

return 0;


updateLong(s, s2, len)

	Описание
	Исправить MEMO поле s  на первые len позиций массива s2

Поле s должно быть в формате ИмяТаблицы/ИмяПоляMEMO/ИмяПоляИД/ЗначениеЭтогоПоля

	Пример
	var prog[3000], s[14], s_ins[14];

sput(prog, "put(\"Нет программы у вида оплаты с кодом %d\", Код_ВО()); return 0;");

l = strlen(prog);

sput(s, "sl_pay/prog/id/%s", 12);

sput(s_ins, "sl_payc/prog/id/%s", 12);

updateLong(s, prog, l);

zcompileDB(s, s_ins, 4);

return 0;


DO_SQLFUNC(строка_результата, строка_вызова_функции)

	Описание
	Выполняет серверную функцию (функцию из пакета) и результат в виде строки возвращает в строка_результата

 DO_SQLFUNC(result, "sl_a.w_days(200, NULL, NULL, 2000* 12+ 5, 2, 'WDAYS')");




SelectInto(SQL_stmt, rezult1, rezult2, ...);

	Описание
	В переменные rezult1, rezult2, ... записываются результаты

исполнения select-предложения SQL_stmt, если запрос возвращает

хотя бы одну строку. Иначе переменные rezult1, rezult2, ...

остаются без изменений. Если запрос выдает более одной строки,

читается только первая. Количество переменных rezulti должно

быть меньше или равно количеству полей в select-предложения SQL_stmt.

Причем, для числовых полей, результат также записывается в числовые 

переменные, а для остальных в строки

Функция SelectInto(...) эквивалентна примерно следующему  

open t SQL_stmt;

fetch t, srezult1, srezult2, ...;

if (!teot(t)) {

if (тип столбца числовой)  strnum(srezult1, rezult1); 

... 

if (тип столбца числовой)  strnum(srezult1, rezult1); 

}

close t;



	Возвращаемое значение
	Возвращает 0 при успехе и не 0 иначе

	Пример
	Пример
   var t, t2, s[2], sl[200],id, p_ord;

   if (SelectInto("select id, sname, name, pay_ord from sl_pay where id = 25",

         id, sn, sl, p_ord))

      put("Нет ВО с кодом id = %d", id);

   else  

      put("id = %d  pay_ord = %d  sname = %s, name = %s",

        id, p_ord, sn, sl);




Пример программы с использованием SQL операторов:

var FirmID, PDate[2], t, tch, PeopNum, 

    ChildsNum[2], chn, i, PeopID, PeopName[5], str[2];

FirmID = 2;
// Maximum children

strcpy( PDate, "01/01/1900" );

Open t Select distinct p.pId, c.Name, p.id_Firm, ch.pID 


     from people p, card c, children ch


where p.auto_card = c.auto_card AND


      p.id_Firm = :FirmID AND


      ch.pID = p.pID AND


      ch.Date_comp > :PDate;

PeopNum = tcount( t );

i = 0;

put( "Where are %d people who need compensation", PeopNum );

while ( i < PeopNum )  // PeopNum - For each selected record

{  Fetch t, str, PeopName;

   strnum( str, PeopID );

   Open tch Select count(*) from children



where pId = :PeopID AND


      
      Date_comp > :PDate;

   Fetch tch, ChildsNum;

   strnum( ChildsNum, chn );

   put( "%s has %s child(ren).", PeopName,  chn );

   Close tch;

   i = i + 1;

}

CloseSelect t;

return 0;

Функции ввода/вывода  с файлами

Файлы бывают двух типов. Текстовые – числа представлены в виде строк, переводы строк записываются двумя знаками (\n\r). Двоичные – числа записываются во внутреннем представлении, переводы строк (новые строки) разделяются одним символом. Точнее в двоичном файле вообще нет никаких строк – есть только поток байтов.

   Во всех операциях с файлами, если имя файла начинается с ">", то файл открывается для чтения или записи в каталоге, возвращаемом функцией 

     GetHostVariable("GLOBAL.docpath", DocDirectory);

Знак ">" естественно отбрасывается

f = fopen(file_name, flag)

	Описание
	 -открыть файл file_name по режиму flag
Оба параметра строки (массивы)

file_name - имя файла по правилам операционной системы

Допустимые значения flag  “r” -для чтения , “w” -для записи, “a” - для добавления к концу файла. Если flag “dr”, “dw”, “da”, то предполагается, что операция чтения и записи в файл будут производиться в кодировке Dos (866)

Если к флагу добаляется буква “b” (т.е. “rb” , “wb”, “ab”, “drb” , “dwb”, “dab”), то файл открывается для чтения или записи в двоичном виде и значит можно пользоваться функциями fread_num, fread_str, fwrite_num, fwrite_str.

Если возвращаемое значение < 0, то файл не открыт.


n = fget(f, buf, max_read);

	Описание
	Прочитать из файла f строку в массив buf  не более чем max_read символов.

Символ перевода строки при этом отсекается.     При конце файла возвращаемое значение 0.


fput(f, format_str, p1, p2, …);

	Описание
	Запись в файл f строки, сформированной из параметров p1,p2,… по строке    форматирования format_str.  Аналогично функции sput, однако вывод в файл


fread_num(number f, number t, &val);

	Описание
	 чтение из файла f значения в переменную val как

    целое размером в 1 байт при t = 1

    целое размером в 2 байт при t = 2

    целое размером в 4 байт при t = 4

    вещественное размером в 4 байт при t = 14

    вещественное размером в 8 байт при t = 18

    символ при t = 21 ('я' прочитается как 255 с возможной перекодировкой

если файл был открыт с параметром d)


fread_str(number f, number t, string s[], n);

	Описание
	 чтение n символов из файла f в строку s
    t = 22 0 в конце не добавлять (строка будет неверной)

    t = 23 в конце добавить 0


fwrite_num(number f, number t, val);

	Описание
	запись в файл  f значения val как

    целое размером в 1 байт при t = 1 (255 запишется как -1)

    целое размером в 2 байт при t = 2

    целое размером в 4 байт при t = 4

    вещественное размером в 4 байт при t = 14

    вещественное размером в 8 байт при t = 18

    символ при t = 21 (val = 255 запишется как 'я' с возможной перекодировкой

          если файл был открыт с параметром d)


fwrite_str(number f, string s[], n);

	Описание
	 запись в файл f строки s, но не более n символов 

  0 в конце не добавляется


fclose(f)  

	Описание
	Закрытие файла

	Пример
	копирования файла f1.t в файл f1_cpy.t

var s[200], f_inp, f_out;

f_inp = fopen(“f1.t”, “r”);

f_out = fopen(“f1_cpy.t”, “w”);

if (f_inp < 0 || f_out < 0) {

    put(“Невозможно открыть файл f1.t  или f1_cpy.t”);  return 0;

}

while(1) {

     if (fget(f_inp, s, 1500) <= 0) break;

     fput(f_out, “%s\n”, s);

}

fclose(f_inp);

fclose(f_out);

return 0;


dput(file_name, format_str, p1, p2, ...);

	Описание
	Запись в файл file_name строки сформированной из параметров p1,p2,… по строке

форматирования format_str путем добавления к концу файла. Аналогично функции sput, однако вывод в файл

Если file_name = "nul", то вывод в файл не производится.

Если в результате форматирования получившаяся строка будет равна "!!DropFile", то   файл file_name будет открыт на запись с параметром "w" 

(т.е будет удален и создан заново)


GetCurDir(file);

	Описание
	Получить текущий каталог в строке file


n = GetOpenFileName(file)

	Описание
	Вызвать диалог для открытия файла для чтения. Если n не ноль, то пользователь отказался от выбора файла


n = GetSaveFileName (file)

	Описание
	Вызвать диалог для открытия файла для записи. Если n не ноль, то пользователь отказался от выбора файла


CopyFile(старый_файл, новый_файл, признак_перезаписи);

	Описание
	Копировать «старый файл» в «новый_файл». Если «признак_перезаписи» = нулю, то файл не будет скопирован, если «новый файл» уже существует

Если не ноль, копировать с перезаписью


MoveFile(старый_файл, новый_файл);

	Описание
	Переименовать файл «старый_файл» в «новый файл».


DeleteFile(file);

	Описание
	Удаление файла file

	Пример
	var file[40], file_new[40];

GetCurDir(file);

put("GetCurDir = %s", file);

strcpy(file, "*.txt");

if (GetOpenFileName(file))

   put("Error for GetOpenFileName(...)");

else   

   put("GetOpenFileName(...) = %s", file);

strcpy(file_new, "*.txt");

if (GetSaveFileName(file_new))

   put("Error for GetSaveFileName(...)");

else   

   put("GetSaveFileName(...) = %s", file_new);

if (strlen(file_new))

   put("CopyFile('%s', '%s', 1) = %d", file, file_new, CopyFile(file, file_new, 1));

if (strlen(file_new))

   put("MoveFile('%s', '0045.667') = %d", file_new, MoveFile(file_new, "0045.667"));

strcpy(file_new, "0045.667");

put("DeleteFile(%s) = %d", file_new, DeleteFile(file_new));

return 0;


LoadFileFromClient(string файл_на _сервере[], string файл_на_клиенте[]);

	Описание
	Загрузить с локального компьютера «файл_на_клиенте» на сервер приложений в файл «файл_на_сервере». Только для версии Oracle.


SaveFileToClient(string файл_на _сервере[], string файл_на_клиенте[]);

	Описание
	Загрузить на локальный компьютер в «файл_на_клиенте» с сервера приложений  «файл_на_сервере». 

Только для версии Oracle.


Z_filename(number файл, string имя_файла[]);

	Описание
	Вернуть имя файла имя_файла для открытого файла с номером файл. 




n = FileExists(string имяфайла[]  [, number &file_size [, string &file_time[] ]]);

	Описание
	Возвращает 1 – если существует файл имяфайла или 0 – иначе. Если указаны параметры file_size  и file_time, то функция также вернет размер файла в байтах и время создания в формате dd/mm/yyyy hh24:mi:ss.

В версии Oracle файл ищется на сервере приложений.


Srv_UserDir(string каталог[]);

	Описание
	Возвращает в строку каталог имя каталога для файлов текущего пользователя.

Тождественно GetHostVariable("GLOBAL.docpath", каталог);




client_UserDir(string каталог[]);

	Описание
	Возвращает в строку каталог имя каталога для файлов текущего пользователя на компьютере пользователя.

Для версии SQL тождественно Srv_UserDir(каталог)



client_FileExists(string имяфайла[]  [, number &file_size [, string &file_time[] ]]);

	Описание
	Возвращает 1- если на компьютере текущего пользователя существует файл имяфайла или 0 – иначе. В остальном аналогична функции FileExists(….).

Для версии SQL тождественно FileExists(имяфайла, file_size, file_time);


client_GetCurDir(file);

	Описание
	Получить текущий каталог в строке file на компьютере пользователя
Для версии SQL тождественно GetCurDir(file)


client_CopyFile(старый_файл, новый_файл, признак_перезаписи);
	Описание
	Копировать «старый файл» в «новый_файл» на компьютере текущего пользователя. Если «признак_перезаписи» = нулю, то файл не будет скопирован, если «новый файл» уже существует.

Если не ноль, копировать с перезаписью.

Для версии SQL тождественно CopyFile(старый_файл, новый_файл, признак_перезаписи);


client_MoveFile(старый_файл, новый_файл);

	Описание
	Переименовать файл «старый_файл» в «новый файл» на компьютере текущего пользователя.

Для версии SQL тождественно MoveFile(старый_файл, новый_файл);


client_DeleteFile(file);

	Описание
	Удаление файла file на компьютере текущего пользователя.

Для версии SQL тождественно DeleteFile(file);


Функции для создания отчетов

ropen(“Имя_Отчета”, “”)

	Описание
	Начинает создание отчета с именем “Имя_Отчета”


rclose()

	Описание
	Заканчивает создание отчета и отображает его на экране 


rput(“ Строка формата ”, p1, p2 …)

	Описание
	Форматированный вывод информации в отчет;

p1, p2 - параметры. Формат команды соответствует команде put.


WrapString(str, len, s1, s2, s3 …)

	Описание
	Разбиение текстовой строки str на подстроки s1, s2 … длиной не более len с выравниванием.


NumMask(x, mask, str)

	Описание
	Преобразование числа x в строку str по маске mask.


NumAlpha(x, len, s1, s2 …)

	Описание
	Преобразование числа x в строковое представление с разбиением на несколько строк p1, p2 длиной не более len.


ДнейМесяца(mm, yy)

	Описание
	Дней в  месяце mm.


НМесяц(mm, str, p)

	Описание
	Преобразование месяца по номеру mm в строку str.

p = 0 для именительного падежа

p = 1 для родительного падежа.


Пример использования данных функций

var st[70], p1[10], p2[10], p3[10];

//put("тест отчета");

r = ropen("zrp_r_tst", ""); 

var st1[3], st2[3];

rput(" Тестовый отчет на PREVIEW\r");

rput("                 стр. %2d\r", 1);

rput("----------------------------\r"

     "     е р у н д а            \r"

     "\r\r\r"

     "           Петров К.Н.\r");

a = 1234567.89;

n = numAlpha(a, 40, p1, p2, p3);

rput(" число %d\r"

     " и в символьном виде\r"

     "%s\r%s\r%s\r-------------\r",

       a, p1, p2, p3);

strcpy(p1, "ZZZ'ZZZ'ZZZ.ZZ");

numMask(a, p1, st);

rput(" число %d\r"

     " по маске %s  ---    %s\r"

     "-------------\r",

       a, p1, st);

strcpy(st, "Строка примерно длинновата, но не так чтобы очень, всего символов 80");

wrapString(st, 40, p1, p2, p3);

rput("строка \r%s\r"

     "разбитая по 40 знаков\r"

     "%s\r%s\r%s\r"

     "-------------\r",

       st, p1, p2, p3);

rclose();

return 720;

Создание отчетов в RTF-формате

Если во втором аргументе функции ropen указать "rtf", то отчет будет

подготовлен в формате RTF

Другая возможность - будет распространяться на все отчеты, установитьконстанту программы sl_rput_type в значение RTF. В этом случае для

формирования обычных текстовых отчетов надо второй аргумент установить

равным "txt".

Строка форматов функции rput пополнилась следуюшими кодами

\n - перевод строки

\12 - перевод страницы

Остальные коды начинаются с последовательности \8

\8b - включить/выключить полужирный фонт

\8i - включить/выключить курсив

\8u - включить/выключить подчеркивание

\8s - включить/выключить зачеркивание

\8a - включить фонт Arial Cyr

\8c - включить фонт Courier New Cyr
\8ann - включить фонт Arial Cyr размером nn

\8cnn - включить фонт Courier New Cyr размером nn

\8pl - выровнять параграф влево

\8pr - выровнять параграф вправо

\8pc - выровнять параграф по центру страницы

\8ol - ориентация страницы "Пейзаж"

\8op - ориентация страницы "Портрет"

\8r - весь текст после этой команды и до конца строки будет рассматриваться

      как текст в формате RTF (но русские буквы будут все же перекодированы)

Эта команда позволяет расширить отчет другими возможностями формата RTF

Например, по умолчанию отчеты готовятся для формата A4, а нужен A3.

rput("\8r\\lndscpsxn\\pgwsxn23814\\pghsxn16840\n");

Если невозможно весь текст уместить в одном вызове rput, то тогда можно

сделать несколько вызовов rput вида

rput("\8r.......");

.....

rput("\8r.......");

Здесь вместо точек используются команды для RTF формата

Выделить цветом абзац

rput("\8r\\cf6");

rput("  Этот параграф выделен красным цветом цветом\n\n");

rput("\8r\\cb0\\cf0");

cf1  черный

cf2  синий

cf3  бирюзовый

cf4  ярко-зеленый

cf5  лиловый

cf6  красный цвет

cf7  желтый

cf8  белый

cf9  темно-синий

cf10 сине-зеленый

cf11 зеленый

cf12 фиолетовый

cf13 темно-красный

cf14 коричнево-зеленый

cf15 серый 50%

cf16 серый 25%

Если после команды в отчете сразу идут цифры, 

рекомендуется эти команды писать на отдельной строке, т.к. иначе все 

последующие цифры будут восприняты как параметры этой команды, например,

как размер 456 фонта Courier для команды \8c456

Второе ограничение - вызов функции rput должен целиком содержать всю команду

То есть следующая последовательность ошибочна

rput("\8"); rput("bТест RTF\8"); rput("b\n");

Пример

// тест RTF отчета

var st[70], p1[10], p2[10], p3[10];

r = ropen("rtf_tst", "rtf"); 

rput("\8ol");

rput("\8b\8iТест RTF\8i\8b\n");

rput("\8a14");

rput("Строка со шрифтом Arial\n");

rput("\8c12");

rput("Строка со шрифтом Courier\n");

rput(" \n\n\n\n Пропущено 4 строки\n");

rput(" \8uСтрока со шрифтом Courier подчеркнутая\8u\n");

rput("\8a14");

rput("\8sСтрока со шрифтом Arial перечеркнутая\8s\n");

rput(" \8bСтрока со шрифтом Arial жирная\8b\n");

rput(" \8iСтрока со шрифтом Arial курсив\8i\n");

rput("\12");  // новая страница

rput("\8op");

rput("\8pcСтраница 2\n\n");

rput("\8pl");

rput("  Этот параграф с очень длинной строкой (специально подготовленной для данного теста с целью максимальной проверки возможностей функции rput) выровнен влево\n\n");

rput("\8pr");

rput("  Этот параграф выровнен вправо\n\n");

rput("\8pc");

rput("  Этот параграф выровнен по центру страницы\n\n");

rclose();

return 702;

// Тест цветов
var st[70], p1[10], p2[10], p3[10];

r = ropen("zrp_r_tst", "rtf"); 

rput("Тест RTF\n");

rput("\8r\\cf1");

rput(" cf1 код цвета\n");

rput("\8r\\cf2");

rput(" cf2 код цвета\n");

rput("\8r\\cf3");

rput(" cf3 код цвета\n");

rput("\8r\\cf4");

rput(" cf4 код цвета\n");

rput("\8r\\cf5");

rput(" cf5 код цвета\n");

rput("\8r\\cf6");

rput(" cf6 код цвета\n");

rput("\8r\\cf7");

rput(" cf7 код цвета\n");

rput("\8r\\cb1\\cf8");

rput(" cf8 код цвета\n");

rput("\8r\\cf9");

rput(" cf9 код цвета\n");

rput("\8r\\cf10");

rput(" cf10 код цвета\n");

rput("\8r\\cf11");

rput(" cf11 код цвета\n");

rput("\8r\\cf12");

rput(" cf12 код цвета\n");

rput("\8r\\cf13");

rput(" cf13 код цвета\n");

rput("\8r\\cf14");

rput(" cf14 код цвета\n");

rput("\8r\\cf15");

rput(" cf15 код цвета\n");

rput("\8r\\cf16");

rput(" cf16 код цвета\n");

// восстановить цвет по умолчанию

rput("\8r\\cb0\\cf0");

rclose();

return 702;

Создание отчетов для Microsoft Office
Доступно только для версии Oracle.
Формирование таких отчетов происходит в два этапа. Сначала Z-программа записывает данные в таблицы базы данных. Затем, уже после окончания работы Z-программы, с помощью стандартных средств Oracle Forms осуществляется реальный вывод в документы Word или Excel. Подробнее смотрите в документе «Настройка_программы.doc». С версии обновления 080616 рекомендуется использовать Z-модули  m_Excel и m_Word, значительно облегчающие написание отчетов.
ins_Excel(sid, НОМЕР_СТРОКИ, НОМЕР_СТОЛБЦА, ПОДСТАВЛЯЕМОЕ_ЗНАЧЕНИЕ);

	Описание
	В версии Oracle выполняет
insert into hr_exp_excel(s_id, n, r, c, s1, isnum)

  values(sid, n1, НОМЕР_СТРОКИ,НОМЕР_СТОЛБЦА,

            ПОДСТАВЛЯЕМОЕ_ЗНАЧЕНИЕ', 0)

В версии SQL выполняет

insert into pr_rep(s_id, n, f1, f2, s1, s2)

  values(sid, ВНУТР_НОМЕР, НОМЕР_СТРОКИ, НОМЕР_СТОЛБЦА, 'ПОДСТАВЛЯЕМОЕ_ЗНАЧЕНИЕ', '')


ins_Excel_num(sid, НОМЕР_СТРОКИ, НОМЕР_СТОЛБЦА, ПОДСТАВЛЯЕМОЕ_ЗНАЧЕНИЕ);

	Описание
	В версии Oracle выполняет
insert into hr_exp_excel(s_id, n, r, c, s1, isnum)

  values(sid, n1, НОМЕР_СТРОКИ,НОМЕР_СТОЛБЦА,

            ПОДСТАВЛЯЕМОЕ_ЗНАЧЕНИЕ', 1)

В версии SQL выполняет

insert into pr_rep(s_id, n, f1, f2, s1, s2)

  values(sid, ВНУТР_НОМЕР, НОМЕР_СТРОКИ, НОМЕР_СТОЛБЦА, 'ПОДСТАВЛЯЕМОЕ_ЗНАЧЕНИЕ', '1')


ins_Word(sid, ЗАКЛАДКА, ЗНАЧЕНИЕ);

	Описание
	В версии Oracle выполняет
insert into hr_exp_word(s_id, n, s0, s1)

   values(sid, n1, ЗАКЛАДКА, ЗНАЧЕНИЕ);

В версии SQL выполняет
insert into pr_rep(s_id, n, s0, s1)

  values(sid, n1, ЗАКЛАДКА, ЗНАЧЕНИЕ)


Пример
import m_Excel;

{

  string sid[15];

  number t;

  open t select SESSIONID from dual;

  fetch t, sid;

  close t;

  m_Excel.Init(sid);

  if (get("Использовать клиентский Excel")) {

     m_Excel.set_mode(2);   //ins_Excel(s_id, 0, 0, "__OLE_MODE__2");

  }

  m_Excel.str(1, 1, "Это");  //ins_Excel(s_id, 1, 1, "Это");

  m_Excel.str(2, 1, "российский");  m_Excel.ColorCell(2, 1, 5);

  m_Excel.str(3, 1, "флаг"); m_Excel.ColorCell(3, 1, 3);

  m_Excel.WidthColumn(1, 50);

  m_Excel.HeightRow(1, 48);

  m_Excel.HeightRow(2, 48);

  m_Excel.HeightRow(3, 48);

  m_Excel.num(1, 3, 12);  //ins_Excel_num(s_id, 1, 1, "12");

  m_Excel.num(2, 3, 6);

  m_Excel.num(3, 3, 1991);

  m_Excel.SelectCell(3, 3);

  return 0;

}

Функции ввода/вывода на экран дисплея


get(формат,пер1,...)

	Описание
	Функция с переменным числом аргументов.

Ввод осуществляется согласно строке формат в переменные пер1, пер2,...

Строка формата представляет собой комбинацию обычных символов и спецзнаков %.

Эталонный формат:

 %d - означает ввод числа,

%s - означает ввод строки,

и, естественно, этому аргументу должен быть сопоставлен массив.

Может быть указана длина вводимых значений, например: %3d, %10s

days = 10;

get("введите количество дней %2d",days);

на экране появится окно вида:

При этом можно ввести только число не более двух знаков или же нажать <ОТКАЗ>.

Функция возвращает количество введенных полей или 0, если нажать кнопку <ОТКАЗ>.

Для Windows клиента возможно только одно вводимое поле


put(формат,пер1,...)

	Описание
	Вывод на экран, где строка формат точно такая же, как и в функции sput.

В строке формат можно использовать символ "\n", что означает переход на новую строку.

put(ввели \n фамилия %79 \n оклад %d", name,wage);


Набор функций Console…  позволяет организовать консольное окно

вывода, аналогичное тому, которое показывается во время расчета зарплаты.

c =  ConsoleOpen(Заголовок, Ширина_бегущей_строки)

   Открыть консольное окно с заголовком Заголовок   и с шириной ползунка             Ширина_бегущей_строки. Ширина устанавливается в относительных величинах, например – это количество строк запроса. Текущее заполнение ползунка – это отношение, например, уже прочитанных строк к общему количеству строк в запросе

 Возвращает номер консоли (0..2) или <0 при ошибке

break= ConsolePut(c, строка, позиция)

    Послать строку строка на консоль c  с текущей шириной заполнения ползунка равной позиция
    Возвращает 1 при успехе и 0 если пользователь нажал CANCEL

ConsoleClose(c)

    Закрыть console c
ConsoleSetProgressBar(c, w)

   Установить линейку ползунка шириной в w значений

ConsoleSetButtonCancel(c, but)

   Погасить кнопку Отказ если but = 0, и восстановить иначе

c =  ConsoleOpen(Заголовок, Ширина_бегущей_строки)

   эквивалентно последовательности

  c =  ConsoleOpen(Заголовок, 0)

  ConsoleSetProgressBar(c, Ширина_бегущей_строки)

  Пример:

  var tabn[1], fname[10], ss[20];

  open t

    select count(*) from sl_emp where s_id = sessionid;

  fetch t, ss;

  close t;

  strnum(ss, t);

  c =  ConsoleOpen("Список работников", t);

  //c =  ConsoleOpen("Список работников", 0);

  //ConsoleSetProgressBar(c, t);

  open t

    select tab_n, fname from sl_emp where s_id = sessionid order by 2;

  ConsoleSetButtonCancel(c, 0);

  i = 1;

  while (1) {

     fetch t, tabn, ss;

     if (teot(t)) break;

     if (i == 10)

        ConsoleSetButtonCancel(c, 1);

     sput(ss, "%4d. %7s  %s", i, tabn, fname);

     if (ConsolePut(c, ss, i) <= 0) break;

     i = i+1;

  }

  close t;

  ConsoleClose(c);

  return 0;

      put_to_Console(t)

  если t = 1 то функция put выводит свои сообщения

  не в отдельное окошко, а в основное (окно отображения расчета зарплаты)

  параметр t = 0 возвращает вывод в нормальное состояние

Пример:

  var ss[20];

  put_to_Console(1);

  x = 0;

  cind = getТекВО();

  i = 1;

  while (1) {

     cp = Код_ВО(i);

     if (cp == 0) break;

     setТекВО(i- 1);

     put("ВО = %d мес = %d Сумма = %d", 

            Код_ВО(), Месяц_ВО(),  Сумма_ВО());

     i = i+1;

  }

  setТекВО(cind);

  put_to_Console(0);

   с = getConsoleMain()

        Получить номер основной консоли – основного окна отображения расчета зарплаты

Взаимодействие Z-программ

zexec (“Имя”, p1, p2 …)

zexecс (“Имя”, cx, p1, p2 …)

zexecs (s,  p1, p2 …)

zсompileDB (s1, s2, t)

	Описание
	Выполнить другую Z программу с именем “Имя”.

p1, p2  - строковые параметры, передаваемые в программу.

zexecc выполнить некомпилированую Z-программу, выделив для компиляции буфер размером cx  байт

zexecs выполнить некомпилированую Z-программу, находящуюся  в строке s

zcompileDB выполнить компиляцию Z-программы s1 и записать откомпилированный код в s2 Параметр t  если рассматривать его как двоичное число означает

b1b2b3                      bi 0 или 1

b1 - записывать в БД

b2 - выдать информацию о компиляции

b3 - выдать листинг если возможно

Параметры s1, s2, Имя имеют следующий формат 

ИмяТаблицы/ИмяПоляMEMO/ИмяПоляИД/ЗначениеЭтогоПоля

Если это z-скрипт то формат упрощенный просто Имя z-скрипта Если же Z-программа находится в другой таблице то ее имя должно следовать полному формату. 

	Пример
	sl_pay/prog/id/20


ParamCount()

	Описание
	Возвращает количество параметров, переданных в программу при вызове оператором zexec


ParamStr(n, str)

	Описание
	Получить в строку str значение n-го параметра, переданного в программу. Параметры нумеруются с 0


PutGlobArray(n, array, nn)

	Описание
	Записать содержимое массива внутри программы в глобальный (доступный другим программам) массив;

n - номер глобального массива,  n - 0..19;

array - имя массива внутри программы;

nn - количество передаваемых элементов в массиве.


GetGlobArray(n, array, nn)

	Описание
	Получить содержимое глобального массива в массив в программе

n - номер глобального массива,  n - 0..19;

array - имя массива внутри программы;

nn - количество читаемых элементов в массиве.


FreeGlobArray(n)

	Описание
	Освободить память из под глобального массива. ,  n - 0..19


ГлобПараметры(мас)

	Описание
	Z-Программы также могут обмениваться с другими  Z-программами через массив из 20 элементов. Данная функция получает значение этих глобальных параметров. Перед каждым началом расчета ЗП массив обнуляется.


SetГлобПараметры(мас)

	Описание
	Установить значение глобальных параметров.


С момента вызова User_Exit (Oracle) или первого вызова udf (Ms SQL) создается автоматически управляемая глобальная память в виде двух динамически расширяемых массивов.

По окончании автоматически удаляется. Глобальная память выделяется блоками

по 512 элементов при записи в нее функциями GMput, GMputArr, UGMput, UGMputArr.

Чтение из памяти по адресу, для которого еще не выделен блок, приводит к ошибке

Первый массив (функции GM…) предназначен для использования разработчиками «БОСС-Кадровика», второй (функции UGM…) для доработок клиентами.

GMput(adr, value)

UGMput(adr, value)
	Описание
	записать значение value по адресу adr


value = GMget(adr)

value = UGMget(adr)
	Описание
	прочитать в value значение глоб. памяти по адресу adr


GMputArr(adr, array[], len)

UGMputArr(adr, array[], len)

	Описание
	 записать массив array длиною len начиная с адреса adr

 тождественно
 i = 0;  while (i < n) { GMput(adr+ i, array[i]);  i= i+ 1; }


GMgetArr(adr, array[], len)

UGMgetArr(adr, array[], len)

	Описание
	   прочитать len элементов глоб. памяти с адреса adr в массив array

   тождественно
   i = 0;  while (i < n) { array[i] = GMget(adr+ i);  i= i+ 1; }


{

  number n;

  var dd1[3], dd2[3];

  string str1[100];

  string f[100];

  put("BEGIN\n");

  GMput(10, 13);  GMput(1000, 9);  GMput(10000, 1962);

  f = "Да это же мой день";

  GMputArr(3000, f, sizeof(f));

  put("READING\n");

  dd1[0] = GMget(10);  dd1[1] = GMget(1000);  dd1[2] = GMget(10000);

  GMgetArr(3000, str1, sizeof(f));

  put("dd1 = %d.%d.%d -- %s\n", dd1[0], dd1[1], dd1[2], str1);

  put("UGM BEGIN\n");

  UGMput(10, 16);  UGMput(1000, 2);  UGMput(10000, 1966);

  f = "Нет, это не мой день";

  UGMputArr(3000, f, sizeof(f));

  put("READING\n");

  dd1[0] = UGMget(10);  dd1[1] = UGMget(1000);  dd1[2] = UGMget(10000);

  UGMgetArr(3000, str1, sizeof(f));

  put("dd1 = %d.%d.%d -- %s\n", dd1[0], dd1[1], dd1[2], str1);

  put("GM ERROR\n");

  n = GMget(30000);

  put("err -- %d\n", n);

  put("UGM ERROR\n");

  UGMgetArr(30000, str1, sizeof(f));

  put("err -- %s\n", str1);

  return 0;

}

Для взаимодействия Z программ и запускающей программы(которой и является БОСС-Кадровик) прдназначены функции

GetHostVariable(string name, string &value)

GetHostVariable_num(string name, number &value)

SetHostVariable(string name, string &value)

SetHostVariable_num(string name, number value)

GetHostVariable(string name, string &value)

GetHostVariable_num(string name, number &value)

	Описание
	Прочитать значение поля name в переменную value


SetHostVariable(string name, string &value)

SetHostVariable_num(string name, number value)

	Описание
	Установить значение поля name из переменной value


Если функции возвращают 0, операция успешна,

 иначе же значение переменной value не изменяется (в функциях SetHost…)

Функции предназначены для чтения и установки значений полей текущей формы и некоторых

глобальных переменных

   Для Oracle name  ИмяБлока.ИмяПеременной

   Для SQL    name  ИмяСписка.ИмяПоля

Имена GLOBAL.docpath, GLOBAL.filepath, GLOBAL.cachepath используются в след. контексте

   GetHostVariable("GLOBAL.docpath", DocDirectory);

   прочитать в DocDirectory каталог для документов

   В версии Oracle возвращается каталог, указанный пользователем для документов
   на сервере приложений.
   В версии Ms SQL возвращается путь к папке документов в профиле пользователя.

   GetHostVariable("GLOBAL.tmppath", FileDirectory); 

    прочитать в FileDirectory каталог для временных файлов
   Возвращается каталог, на который указывает переменая окружения TEMP

   В версии Ms SQL возвращается путь к «временному» (для хранения «рабочих» файлов) 

   каталогу  Windows.

   GetHostVariable("GLOBAL.cachepath", DocDirectory); 

    прочитать в DocDirectory каталог для шаблонов отчетов

   В версии Oracle равносильно GetHostVariable("GLOBAL.filepath", DocDirectory);

   В версии Ms SQL возвращается путь к Cache-каталогу в профиле пользователя. 
ЯЗЫК Z 2





(редакция для платформы RP Server)

